

	A	B	C	D	E	F	G	H
1	Umrechnung der Parameter der GF-Verteilung für Unterverteilungen							
2								
3	Verteilung		GF-Verteilung	Log-logistische Verteilung	Log-Normal-Verteilung	Weibull-Verteilung	Gamma-Verteilung	
4								
5	@RISK-NAME			RISK LOGLOGISTIC	RISK LOGNORM2	RISK WEIBULL	RISK GAMMA	
6								
7			Parameter der GF-Verteilung (rote Angaben sind fest)					
8	Parameter							
9		m1=	9.11183	1	∞	1	11.74583	
10		m2=	9.70326	1	∞	∞	∞	
11		λ=	0.00049	0.00049	0.00049	0.00042	0.00047	
12		p=	1.61247	5.95789	3.40278	3.36433	1	
13								
14				Parameter der @RISK-Verteilungen				
15	Parameter							
16	<i>Shape</i>	α=		5.9579		3.3643	11.7458	
17	<i>Scale</i>	β=		2042.9009		2363.5074	181.4506	
18	<i>Mean</i>	μ=			7.6213			
19	<i>Std-Dev.</i>	σ=			0.2939			
20	(Umrechnung für Lognormalverteilung)	ω2=			1.0902			
21								
22	@RISK-Befehl			2140.7343	2131.3079	2122.1800	2131.2873	
23								
24			Kennzahlen					
25								
26	Arithm. Mittel	E(X)	2131.0145	2140.7343	2131.3079	2122.1800	2131.2873	
27								
28			Funktionen					
29								
30	<i>Beobachtung</i>	x=	2000.0000	2000.0000	2000.0000	2000.0000	2000.0000	
31	Dichtefunktion	f(x)=		0.0007	0.0007	0.0005	0.0007	
32	Verteilungsfunktion	F(X)=	0.4711	0.4684	0.4723	0.4346	0.4532	
33								
34	<i>Wahrscheinlichkeit</i>	p	0.5	0.5	0.5	0.5	0.5	
35	Quantil	q(p)=	2042.8843	2042.9009	2041.2329	2119.5542	2071.1185	
36								
37								

	A	B	C	D	E	F	G	H
38	Berechnungen:	Zelle =Formel						
39								
40	GF-Verteilung:							
41	Eingaben:							
42	1. Parameter: m1	C9						
43	2. Parameter: m2	C10						
44	3. Parameter: λ	C11						
45	4. Parameter: p	C12						
46	Beobachtung: x	C30						
47	Wahrscheinlichkeit: p	C34						
48	Ausgaben:							
49	Arithm. Mittelwert: E(X)	C26=	EXP(LN(\$C\$10/\$C\$9)/\$C\$12)*EXP(GAMMALN(\$C\$9+1/\$C\$12))					
50			*EXP(GAMMALN(\$C\$10-1/\$C\$12))/C\$11/EXP(GAMMALN(\$C\$9))/EXP(GAMMALN(\$C\$10))					
51	Verteilungsfunktion: F(x)	C32=	1-FVERT(EXP(LN(\$C\$11*\$C\$30)*\$C\$12);2*\$C\$9;2*\$C\$10)					
52	Quantil: q(p)=	C35=	EXP(LN(FINV(\$C\$34;2*\$C\$9;2*\$C\$10))/C\$12)/C\$11					
53	Beachte: Für große Werte von m1 bzw. m2 sind nicht alle Berechnungen mit EXCEL möglich.							
54								
55	Log-logistische Verteilung:							
56	Eingaben:							
57	1. GF-Parameter: λ	D11						
58	2. GF-Parameter: p	D12						
59	Beobachtung: x	D30						
60	Wahrscheinlichkeit: p	D34						
61	Ausgaben:							
62	1. @RISK-Parameter: α	D16=	\$D\$12					
63	2. @RISK-Parameter: β	D17=	1/\$D\$11					
64	@RISK-Befehl:	D22=	RiskLoglogistic(0;\$D\$17;\$D\$16)					
65	Arithm. Mittelwert: E(X)	D26=	EXP(GAMMALN(1+1/\$D\$12))*EXP(GAMMALN(1-1/\$D\$12))/D\$11					
66	Dichtefunktion: f(x)	D31=	\$D\$11*\$D\$12*EXP(LN(\$D\$11*\$D\$30)*(\$D\$12-1))/(1+EXP(LN(\$D\$11*\$D\$30)*(\$D\$12)))^2					
67	Verteilungsfunktion: F(x)	D32=	EXP(LN(\$D\$11*\$D\$30)*\$D\$12)/(1+EXP(LN(\$D\$11*\$D\$30)*\$D\$12))					
68	Quantil: q(p)=	D35=	EXP(LN(\$D\$34/(1-\$D\$34))/D\$12)/D\$11					
69								
70	Log-Normalverteilung:							
71	Eingaben:							
72	1. GF-Parameter: λ	E11						
73	2. GF-Parameter: p	E12						
74	Beobachtung: x	E30						
75	Wahrscheinlichkeit: p	E34						
76	Ausgaben:							
77	1. @RISK-Parameter: μ	E18=	-LN(\$E\$11)					
78	2. @RISK-Parameter: σ	E19=	1/\$E\$12					
79	Umrechnung: ω^2	E20=	EXP(\$E\$19^2)					
80	@RISK-Befehl:	E22=	RiskLognorm2(\$E\$18;\$E\$19)					
81	Arithm. Mittelwert: E(X)	E26=	EXP(\$E\$18)*WURZEL(\$E\$20)					
82	Dichtefunktion: f(x)	E31=	NORMVERT((LN(\$E\$30)-\$E\$18)/\$E\$19;0;1;0)/\$E\$30/\$E\$19					
83	Verteilungsfunktion: F(x)	E32=	NORMVERT((LN(\$E\$30)-\$E\$18)/\$E\$19;0;1;1)					
84	Quantil: q(p)=	E35=	EXP(\$E\$19*NORMINV(\$E\$34;0;1)+\$E\$18)					
85								
86	Weibull-Verteilung:							
87	Eingaben:							
88	1. GF-Parameter: λ	F11						
89	2. GF-Parameter: p	F12						
90	Beobachtung: x	F30						
91	Wahrscheinlichkeit: p	F34						
92	Ausgaben:							
93	1. @RISK-Parameter: α	F16=	\$F\$12					
94	2. @RISK-Parameter: β	F17=	1/\$F\$11					
95	@RISK-Befehl:	F22=	RiskWeibull(\$F\$16;\$F\$17)					
96	Arithm. Mittelwert: E(X)	F26=	\$F\$17*EXP(GAMMALN((\$F\$16+1)/\$F\$16))					
97	Dichtefunktion: f(x)	F31=	WEIBULL(\$F\$30;\$F\$16;\$F\$17;0)					
98	Verteilungsfunktion: F(x)	F32=	WEIBULL(\$F\$30;\$F\$16;\$F\$17;1)					
99	Quantil: q(p)=	F35=	\$F\$17*EXP(LN(LN(1/(1-\$F\$34)))/\$F\$16)					
100								
101	Gamma-Verteilung:							
102	Eingaben:							
103	1. GF-Parameter: m1	G9						
104	2. GF-Parameter: λ	G11						
105	Beobachtung: x	G30						
106	Wahrscheinlichkeit: p	G34						
107	Ausgaben:							
108	1. @RISK-Parameter: α	G16=	\$G\$9					
109	2. @RISK-Parameter: β	G17=	1/(\$G\$9*\$G\$11)					
110	@RISK-Befehl:	G22=	RiskGamma(\$G\$16;\$G\$17)					
111	Arithm. Mittelwert: E(X)	G26=	\$G\$16*\$G\$17					
112	Dichtefunktion: f(x)	G31=	GAMMAVERT(\$G\$30;\$G\$16;\$G\$17;0)					
113	Verteilungsfunktion: F(x)	G32=	GAMMAVERT(\$G\$30;\$G\$16;\$G\$17;1)					
114	Quantil: q(p)=	G35=	GAMMAINV(\$G\$34;\$G\$16;\$G\$17)					
115	Beachte: Für große Werte von m1 sind nicht alle Berechnungen mit EXCEL möglich.							
116								
117								